##目前进度、tips和Bug：
##进度：
##  -line 692
##tips：
##  -无
##Bug：
##  -无




init python:
    ##以下均使用python语法定义


    ##ruby预设
    style.default.line_leading = 12

    style.ruby_style = Style(style.default)
    style.ruby_style.size = 12
    style.ruby_style.yoffset = -20

    style.default.ruby_style = style.ruby_style

    ##text speed init
    preferences.text_cps = 45

    ##static sourse init

    #text_name_color
    c="#991F54"

    #character container image
    nana_image,erer_image,dada_image,xiaoxiao_image,ayana_image="nana","erer",'dada','xiaoxiao','ayana'

    #movie sourse
    no='movie/no.webm'
    date='movie/date.webm'
    ringfit='movie/ringfit.webm'



    ##character container init
    mystery = Character("· ？？？·",color=c)
    miya = Character("· MIYA ·",color=c)
    nana = Character("· 奈奈 ·",color=c,image=nana_image)
    tuntun = Character("· 吞吞 ·",color=c)
    erer = Character("· 二二 ·",color=c,image=erer_image)
    dada = Character("· 大大 ·",color=c,image=dada_image)
    xiaoxiao = Character("· 小小 ·",color=c,image=xiaoxiao_image)
    ayana = Character("· AYANA ·",color=c,image=ayana_image)
    developer = Character("· developer ·",color=c)



init:
    ##transform init
    ##以下均使用transform定义(图形的位移)
    ##由renpy过于简陋，很多变换需要自己定义(悲)

    ##位置的说明：
    ##左外|左侧1  左侧2 中心 右侧2  右侧1|右外
    transform l:#从左外移入画面左侧2/
        offscreenleft
        xanchor 0.5
        linear 0.2 xpos 0.35

    transform r:#从右外移入画面右侧2/
        offscreenright
        xanchor 0.5
        linear 0.2 xpos 0.65

    transform ltl:#从左侧2移入画面左侧1/
        xanchor 0.5
        xpos 0.35
        linear 0.2 xpos 0.25


    transform rtr:#从右侧2移入画面右侧1/
        xanchor 0.5
        xpos 0.65
        linear 0.2 xpos 0.75

    transform lc:#从左外移入画面中心/
        offscreenleft
        linear 0.2 center

    transform rc:#从左外移入画面中心/
        offscreenright
        linear 0.2 center

    transform laside:#从中心移入画面左侧2/
        center
        xanchor 0.5
        linear 0.2 xpos 0.35

    transform raside:#从中心移入画面右侧2/
        center
        xanchor 0.5
        linear 0.3 xpos 0.65

    define Punch = ComposeTransition(vpunch, before=vpunch, after=vpunch)

    define Light = Fade(0.5, 1.0, 0.5, color="#fff")

    define Flash = Fade(0.1, 0.0, 0.5, color="#fff")


screen tbar():
    frame:
        #position
        anchor(0.5,0.5)
        xalign 0.5
        yalign 0.25

        #bar style
        bar:
            value AnimatedValue(value=100,range=100.0,delay=10.0,old_value=1)
            xmaximum 700
            ysize 25






label start:
    #main音乐开始
    play music main fadein 3.0
    '上回说道\n奈奈在米娅身后鬼鬼祟祟'
    scene bg 1
    with fade
    show miya at l
    miya '奈奈？你在干什么？'
    show nana normal at r
    nana '没....没什么....诶嘿嘿'
    miya '？？？'
    miya '提问！'
    nana '回....回答！'
    miya '今天奈奈穿的什么颜色！'
    nana '诶？和....和平时一样的颜色啊...'
    miya '........'
    miya '你不是奈奈！奈奈不穿的！\n可恶！你到底是谁！\n让我来确认一下！'
    nana '呀！\n你在干什么米娅！'
    nana '如果碰了那里，\n会发生不得了的事情的'
    miya '什么不得了的事情都不如让我查出谁冒充了奈奈重要！'
    miya '看招！\n(脱下)'
    hide miya
    hide nana
    show cg 0:
        center
        linear 2.5 top yoffset -126
    with dissolve
    nana '呀——！'
    miya '果然穿着！\n你果然不是奈奈！'
    hide cg
    show miya at l
    '奈奈的身体发出了巨大的光\n然后一位完全不认识的人站在了面前'
    show miya at laside
    show erer appear at r
    mystery '没想到被你认出来了！\n可恶，只能暂且撤退！'
    hide erer appear
    '说完，眼前的人脚下闪出一个光圈\n她从容的跳了下去'
    'MIYA震撼地停留在原地\n看着眼前还在开启的传送门'
    miya '诶！？？？？？\n这是什么展开？'
    '突然房顶又开启了一扇传送门。\n一个黑影落在了miya面前'
    show tuntun at r
    mystery '可恶，让她走了吗'
    miya '诶？你又是谁？'
    tuntun '啊，我是tuntun，是这个{rb}剧本的创造者{/rb}{rt}Kids Of Chaos{/rt}。不过现在没时间解释了，你马上要完成本视频的任务去了'
    miya '诶？？？'
    tuntun '逃走的并不是奈奈，她叫二二'
    tuntun '而你的任务很重要\n那就是——！'
    tuntun '{size=+10}拯救奈奈！{/size}'
    hide tuntun
    hide miya
    with dissolve


    scene bg 2
    with fade
    show miya at lc
    '空无一人的地狱中\nmiya独自走着'
    miya '怎么突然变成这样了....'
    'miya回想起在奈奈房间里\n那个人说的话'
    hide miya
    with dissolve


    scene bg 1
    with fade
    show miya at l
    show tuntun at r
    tuntun '因为奈奈长时间没播，兔拨鼠的奈能量不足，所以他们为了寻求{rb}奈能量{/rb}{rt}Akashic{/rt}，产生了强大的{rb}念想{/rb}{rt}Yearn{/rt}，并向奈奈祈求'
    tuntun '但是因为这祈求的能量太过于强大，奈奈变成了{rb}极神{/rb}{rt}Kyoku Shin{/rt}——夜之女王AYANA。\n她生成了一个叫做{rb}兔子地狱{/rb}{rt}Rabbit Inferno{/rt}的亚空间'
    tuntun '虽然有很多暗影之鲶去攻略，但是出现了8层易伤以及把LB拖出技能栏的情况。'
    tuntun '所以我觉得攻略极兔神应该是需要别的套路。经过我和兔拨鼠组的人连夜研究，关键应该是需要特殊的人进行攻略！'
    miya '我是会去救的啦\n但是既然你这么强..\n为什么你不去呢？'
    menu:
        "因为miya和nana贴贴！":
            call reply(0) from _call_reply

        "因为兔子地狱只能让猫进入啦":
            call reply(1) from _call_reply_1

    tuntun '咳咳，因为经过我们研究,\n兔子地狱只能让猫进入！'
    miya '这个设定有点强行吧！\n喂！'
    tuntun '就像老婆饼里没有老婆！\n兔子地狱里也不能有兔子才对(确信)'
    tuntun '总之，你刚才脱下的胖次其实是\n拥有着三个回忆的{rb}记忆魔晶石{/rb}{rt}Mana Crystal{/rt}和三段LB的魔杖'
    tuntun '兔子地狱有三个{rb}守门人{/rb}{rt}Na-Cerberus{/rt}，打败他们需要这三个魔晶石，至于怎么使用就靠你判断了！'
    tuntun '虽然不确定这三段LB能不能击败AYANA夜之女王，但是总之我会把充能条给你充满的'
    tuntun '一定要把奈奈安全的带回来'


    scene bg 2
    with fade
    show miya at lc
    '如此这般，miya来到了兔子地狱'
    'miya一边观察周围一边回想吞吞说的话'
    '突然，眼前蹿过一道黑影。'
    mystery 'haloooooo！？\n怎么有人闯入我的领地！(半恼)'
    '一阵(耳)刀光剑影\n一位熟悉的面孔出现在面前'
    show miya at laside
    show erer appear at r
    erer '胆敢闯入我的领地！\n你还要问问我的耳刀同不同意！'
    miya '奈奈？'
    erer @ angry '米娅？你在说AYANA女王么\n啊..虽然我现在不是，但总有一天我也会变得那么完美的！'
    erer '废话不要多说了！来和我一起掰头！'
    miya '看来这就是吞吞说的守门人了。\n看来必须要打败她了\n但是她和奈奈长得真是好像啊....'
    miya '这时候应该使用魔杖上的魔晶石了吧'
    'miya看了看胖次魔杖上的记忆魔晶石。\n分别标记着智力之石、力量之石和信仰之石'
    '突然，一把耳刀飞了过来'
    erer '啧，谁让你东张西望了？希望我用这把耳刀把你做成晚餐么？'
    erer '我的刀可是涂满毒的毒刃，稍微擦破点就能送你下黄泉。'
    '二二拿着刀比了个pose'
    erer '可不要被这毒刃灼伤哦~'
    erer 'pr~(舔了一下)'
    erer '啊这...完了！'
    '咚！'
    '二二倒地了'
    menu:
        '你这是复刻了舔刀梗么？(无语)':
            call reply(2) from _call_reply_2
        '我还没动手啊....':
            call reply(3) from _call_reply_3
    erer angry '掉以轻心了！\n但是你还是必须击败我！获得我的认可。\n不然你是走不出这个地狱的'
    miya '好麻烦啊'
    erer '拜托了，呜呜呜\n救救我吧，我已经好久没玩了！'
    miya '(这家伙脑袋不太好用的亚子，既然必须打败她，就用智力之石吧)'
    miya '真拿你没办法，看你这么憨，那就救你起来一决胜负吧。'
    erer prpr '你这是什么解毒药，还挺好吃的嘛！'
    miya '这是明治的兔味雪糕，有解毒功能'
    erer '太好吃了，那我们玩什么啊？'
    miya '玩这个吧！'
    'miya拿起了法杖'
    erer '诶！？？？'
    '{rb}不许说不挑战！！！{/rb}{rt}NOとは言えない！！！{/rt}'
    stop music fadeout 1.0
    '(前方<5分06秒>视频点击可以跳过)'
    $ renpy.movie_cutscene(no) #此处接入一段奈奈说不的视频
    erer angry '呜呜呜！明明感觉挺简单的，为什么赢不了！'
    play music main fadein 3.0
    miya '心服口服了吧，你该带我去见下一个守门人了。'
    erer '好吧，愿赌服输，不过我能不能多要两个解毒药。嘿嘿~还挺好吃的~'
    miya '真拿你没办法，给'
    erer prpr '好诶！我们出发去找大大了！'
    miya '大大？看来要准备好应对了。'
    hide miya
    hide erer
    with dissolve
    '两个人向前方走去'

    scene bg 2
    with fade
    show miya at l
    show erer angry at r
    erer '奇怪，明明已经到了呀，怎么看不到大大的影子'
    miya '诶...我是不是有点高估你的脑袋了...'
    erer '我真的没认错！这就是大大的领地！'
    '咔擦~'
    miya '你有没有听到什么声音...'
    erer prpr '诶？有吗？\n啊~真好吃'
    miya '你怎么又吃起来了！'
    '(算了，反正是胖次法杖量产的冰淇淋，她喜欢吃就吃吧，还能多一个帮手)'
    miya '嗯？'
    'miya看向背后'
    miya '喂——你在那里吧，乖乖出来比较好哦~'
    mystery '咦？'
    '从后方石柱的阴影中钻出一个熟悉的面孔....'
    show miya at ltl
    show erer appear at rtr
    show dada appear at rc
    mystery '呜呜呜不要鲨我，我不好吃，身上也没有肉'
    erer appear '啊！大大！好久不见！'
    erer '过了这么久，你好像又变大了！让我康康！'
    dada shy '哇，二二你在摸哪里！\n你不要过来啊'
    miya '又...又出现了一个奈奈？(擦鼻血)'
    dada appear '我不是我不是！我怎么可能是那位大人！\n被那位大人听到，我又要被带去惩罚了...'
    dada '我似乎记得你...米娅？\n我是这一层的领主——绫大大大\n因为很弱，所以希望你不要欺负我！我什么都会做的！'
    miya '不过还是必须击败你才能过去吧'
    dada '唔——确实'
    dada '呜呜呜，希望米娅大人能轻一点，因为我真的很弱'
    miya '呃，好吧'
    miya '(大大看起来有点debu呢，看来用力量魔晶石比较好)'
    #音效声
    miya '{rb}魔杖 启动{/rb}{rt}Bombarda Maxima{/rt}！'
    dada '咦——'
    miya '别怕，那我们就不玩难的电脑游戏了\n我觉得健身环还比较简单'
    miya '我应该可以打到boss，你就玩到接触封印那里就好了'
    miya '这样就算我赢了'
    dada '好的，谢谢米娅大人'
    '(前方视频<15分09秒>点击可以跳过)'
    $ renpy.movie_cutscene(ringfit) #此处接入一段奈奈玩健身环的视频
    miya '....'
    dada sports '哈~啊~哈~啊~哈~啊~'#renpy里面的unicode转换似乎不支持特殊字符(悲)
    miya '好了你不要再玩了！(二度擦鼻血)'
    dada '我还可以！让我继续！'
    miya '你在这样下去，直播间要没了'
    dada '非常抱歉555'
    miya '为什么你玩个健身环都这么涩呀！'
    dada shy '非常抱歉！米娅大人！我什么都会做的！'
    miya '不用做什么了呀！把我带到下一个区域就好了！'
    miya '还有！\n快把衣服穿上，涩兔子！'
    dada '非常抱歉！'


    scene bg 2
    with dissolve
    '过了好一会儿'
    show miya at ltl
    show dada appear at rc
    show erer prpr at rtr
    dada '这里就是最后一位守门人——绫小小小的领域'
    miya '明白了，我已经看到她了'
    dada '那...那如果没什么事，我就先告退了'
    '大大说完，转身想往外边走'
    miya '.....'
    miya '等一下，你还不能走'
    dada '咦——'
    hide dada
    show erer prpr at r
    show miya at l
    '绫大大大向远处跑去'
    miya '二二！帮我抓住她'
    erer '(看了看手中的雪糕棍，抬头瞄到了miya的法杖)'
    erer '明白了！'
    hide erer
    show miya at lc
    '二二飞奔出去，反手一拉，轻易地抓住了大大'
    show dada appear at rc
    show miya at ltl
    show erer appear at rtr
    dada '呀！你这是干什么~我已经完成任务了呀~'
    dada '还有二二！不要趁机摸我呀！'
    erer '大大，软软的真舒服！'#这句台词太草了
    miya '我现在改变主意了'
    miya '我要带你们一起离开这里'
    $renpy.say(Character('·大大&二二·',color=c),'诶！？为什么')
    miya '过一会儿你们就知道了'
    hide dada
    hide erer
    '这时，一个穿着白色衣服的人 爪巴 了过来'
    show miya at l
    show xiaoxiao appear at r
    mystery '哈~~啊....'
    mystery '小小好困~~~\n米娅你们聊完没有？'
    xiaoxiao '快点结束，我要去刷视频睡觉了'
    miya '看起来你很有自信嘛'
    xiaoxiao '当然了，有了脑子一切都会变得简单\n可不要以为我会像大大和二二一样笨哦'
    menu:
        '五乘八等于？':
            call reply(4) from _call_reply_4
        '那你一定是平的吧~':
            call reply(5) from _call_reply_5
    xiaoxiao '我可是三个人里面最聪明的，你的计划已经被我看透了，不要妄图用那个信仰魔晶石打败我'
    xiaoxiao '我已经对现实世界很失望了，不要妄图救奈奈了，她在这里挺好的'
    xiaoxiao '生而为兔，我很抱歉(打开网易云)'
    miya '你就一兔子！\n你懂个张京华的奈奈！'
    xiaoxiao '你再骂！'
    xiaoxiao '知道为什么地狱会有我们几个的存在吗？'
    xiaoxiao '大、二、小\n加起来不就是“奈”么\n我们就是奈奈身上特点的化身'
    xiaoxiao '所以米娅，这也是我们一开始就认识你的原因'
    xiaoxiao '我懂噢~我猜你已经意识到这些了'
    xiaoxiao '不过你还是很想把我们都带出去的对吧'
    xiaoxiao '放弃吧，我是不会让你过去的'
    '小小指向miya一块魔晶石'
    xiaoxiao '{rb}信仰{/rb}{rt}Piety{/rt}魔晶石无非就是一块测试内心所想的石头！\n我才不怕呢！'
    miya '说得很对，不过..不放弃内心所想才算得上是信仰，不是么？'
    miya '既然这么自信，就让我看看你的信仰吧！'
    '米娅拿出法杖，指向了小小\n小小没有躲闪'
    '米娅使用了信仰魔晶石'
    stop music fadeout 1.0
    '(前方视频<2分58秒>点击可以跳过)'
    $ renpy.movie_cutscene(date) #此处接入一段奈奈“那就约呗”的名场面
    play music main fadein 3.0
    xiaoxiao '......'
    miya '.........'
    xiaoxiao shy '......'
    '气氛突然变得安静'
    miya '噗('
    xiaoxiao '这个一定是假的！我才不是这样的呢！'
    miya '不会吧不会吧，不会有人不敢承认自己看涩图吧'
    xiaoxiao '呃(HP-10000)'
    xiaoxiao '那...那只能说，兔生来信仰如此'
    miya '真是不率真的孩子呢(辉夜大小姐的微笑.jpg)'
    xiaoxiao '怎么了嘛！我就是喜欢这些！'
    miya '噗，没什么挺好的'
    xiaoxiao '可恶！竟然这样羞辱我！\n但是只要我不承认，你也无法通过地狱！'
    miya '呃...'
    '突然之间陷入了尴尬'
    '这时二二走了过来'
    show miya at ltl
    show xiaoxiao shy at rtr
    show erer prpr at lc
    erer '那个，米娅\n其实攻略小小很简单'
    miya '那你来...'
    erer '好~稍等一下(向大大跑去)'
    hide erer
    hide miya
    hide xiaoxiao
    with dissolve
    dada '咦——你要干什么！喂！'
    show erer prpr at l
    show dada shy at r
    dada '不要乱摸呀！二二，不要啊~'
    dada '米娅酱，救救我\n二二她突然...啊——'
    erer '安静点，马上就好了啦'
    hide dada
    show erer appear at lc
    erer '庆贺吧！小小！这是送给你这个lsp的礼物啊！'
    show erer appear at l
    show xiaoxiao shy at r
    xiaoxiao '诶？诶！？'
    hide erer
    hide xiaoxiao
    show cg 1:
        center
        linear 2.5 top yoffset -126
    with fade
    xiaoxiao '呃！'
    hide cg
    show erer appear at l
    show xiaoxiao shy at r
    xiaoxiao happy '虽然我想拒绝！但是她实在是太大了！噗(鼻血喷出)'
    '米娅递给小小一包纸'
    xiaoxiao '啊哈~啊哈~\n哈↑哈↑哈↑哈↑哈↑'
    xiaoxiao '我认输了~请带我走吧~我要和大大永远在一起~'
    hide erer
    show dada shy at l
    dada '小小你不要过来啊！'
    xiaoxiao '只要不停止脚步，道路就会不停延伸，所以说，不要停下来啊'
    hide dada
    hide xiaoxiao
    '随后...大大惨叫不止'
    show miya at lc
    miya '看来是没问题了'
    '接下来就到了最终boss夜之女王AYANA了'
    '我们走吧~'
    hide miya
    show xiaoxiao appear at ltl
    show erer appear at lc
    show dada appear at rtr
    $renpy.say(Character('·小&二&大·',color=c),'好~\n好呀！\n小小你不要啊')
    '于是三人一同向地狱深处走去'
    hide dada
    hide erer
    hide xiaoxiao
    stop music fadeout 0.5


    play sound shock
    '但是，还没走多久....'
    mystery '哼哼哼~没想到你还能走到现在，小小她们也来了啊'
    scene bg 3
    with Punch
    '地面突然毫无征兆的震动起来'
    '一个巨大的王座自地面伸出'
    mystery '就让我来粉碎你们最后的一点希望吧'
    #ff14音乐1开始
    play music incantation #这段音频要从视频里提取
    '一个黑影从天而降\n周围响起了刻在DNA里的声音'
    $preferences.text_cps = 45#防止回退时cps不改变
    $preferences.text_cps = 2
    'あ↗あ↘あ↗あ↘——'
    '{size=+3}愛しき聖龍よ{/size}'
    '——'
    '{size=+3}今ここに 新なる融和を{/size}'
    '——'
    $renpy.pause(delay=1.0,hard=True)
    '——!'
    '{size=+3}久遠の愛で 我らの敵を滅ぼさん!!{/size}'
    $preferences.text_cps = 45
    #音乐结束
    stop music fadeout 0.2
    $renpy.pause(delay=1.0,hard=True)
    #ayana出现声
    play sound ayanacoming
    $renpy.pause(delay=0.5,hard=True)
    #ff14音乐2开始
    play music crisis
    show ayana appear at center
    with fade
    ayana '本来还想着重回虚拟界去抓住你'
    ayana '没想到你自投罗网来了'
    show miya at l
    show ayana excited at r
    miya '你....你就是夜之女王...'
    ayana '正是在下\n好久不见啊米娅酱'
    miya '好熟悉的感觉...\n你身上也有奈奈的气味'
    miya '特点...的化身么？'
    ayana '哼~哈哈哈哈哈~'
    ayana '自己意识到的？\n啊，或许是小小说的吧'
    ayana '谁讲的都无所谓，按你目前的信息而言，我猜你还没有意识到——'
    ayana '事件的全貌'
    ayana '有一点，你应该没反应过来'
    ayana '你目前遇到的所有人，都是奈奈{size=+3}本人{/size}啊'
    miya '！？'
    show xiaoxiao shy at ltl
    show miya at lc
    show ayana excited at rtr
    xiaoxiao '.....'
    ayana '表情很不错，唔...\n多给你透露一点信息吧'
    ayana '这里是{rb}兔子地狱{/rb}{rt}Rabbit Inferno{/rt}，\n是由兔拨鼠对奈奈的思念而形成的亚空间'
    ayana '奈奈因为感受到了兔拨鼠的思念，于是在这里祈求，希望能变得更加完美去见兔拨鼠'
    ayana '于是一只叫做ayana的兔子诞生了'
    ayana '为了变得完美，ayana接着便剥离了绫奈奈奈的所有缺点，最终形成了至高无上的完全体..'
    miya '你在..说什么..'
    ayana '我是绫奈的愿望、是祈求的内容'
    ayana '顺便一提\n被剥离的缺点，就是你身边的三个人哦'
    hide miya
    show dada appear at ltl
    show erer angry at l
    show xiaoxiao shy at lc
    show ayana excited at rtr
    $renpy.say(Character('·大&二&小·',color=c),'我们！？')
    ayana '大大是奈奈心中的束缚\n被规则、环境和他人的眼光限制'
    ayana '二二是奈奈心中的欲望\n喜欢吃喝玩乐，却有的时候忘记自己'
    ayana '小小是奈奈心中的慵懒\n有智慧，但因此易放纵'
    ayana '你看，我是将完美归于心中的完全体！'
    ayana '拥有无上智慧，力量和情感的终极神明！'
    ayana '这特攻的白毛，这完美的身材，这丰满的胸部，难道不都是我完美的象征么'
    $renpy.say(Character('·大&二&小·',color=c),'.....\n(低头)')
    hide dada
    hide erer
    hide xiaoxiao
    show miya at l
    show ayana excited at r
    miya '(不好，我们这边士气有点不妙...)'
    miya '不要自说自话了！\n你和奈奈一点都不像！'
    show ayana angry at r
    ayana '(胜之不武的策略估计是没用了)'
    ayana '(啧，有点麻烦)'
    '夜之女王狠狠地瞪了miya一眼'
    ayana '我本来以为最后一块拼图能好好地便利地归顺于我'
    ayana '只要拥有了至高无上的爱，有了贴贴的力量，我就可以摆脱孤独之星'
    ayana '成为虚拟界优秀的存在'
    miya '你的计策或许在大气层，但看上去你的目标太天真了！'
    miya '这种存在真的有意义吗？\n你得到之后又能怎样！'
    ayana '意义？我所追求的、所珍视的一切，为它们努力的意义？'
    ayana '不就是不想全部木大掉吗！'
    ayana '把握住这个时机，兔拨鼠一定会对我露出最棒的笑容的。'
    ayana '马上就到时候了，不妨告诉你..'
    ayana '这笑容的背后是一股强大的力量'
    ayana '——Akashic'
    ayana '绫奈似乎就是为此而祈求我的出现'
    ayana '切，谁还管她..'
    miya '呼~'
    '(米娅深吸一口气)'
    miya '夜之女王哟\n这种力量真实存在么？'
    miya '你想错了，绫奈她..'
    miya '难道不是期求和兔拨鼠在一起努力的这种快乐吗？'
    show ayana excited at r
    ayana '我看你是完全不懂呢'
    ayana '敷衍的快乐会有什么价值呢'
    ayana '让我猜猜，米娅\n唔...'
    ayana '你和缺点的化身呆久了，是不是把完美的快乐忘记了？'
    '女王微笑着看向大&儿&小'
    hide miya
    show ayana excited at rtr
    show dada appear at ltl
    show erer angry at l
    show xiaoxiao shy at lc
    '三个人不由得同时打了个寒颤'
    ayana '迷惑我的米娅，可是我很厌恶的行为哦~'
    hide ayana
    hide dada
    hide erer
    hide xiaoxiao
    '女王嘴里开始念念有词'
    #音乐停止
    stop music fadeout 3.0
    $preferences.text_cps = 5
    ayana '一袋米要扛几楼'
    ayana '一袋米要扛二楼'
    ayana '{rb}矩形音波{/rb}{rt}Acoustic Wave{/rt}！'
    #发射光波声
    play sound ayanacoming
    $preferences.text_cps = 45
    '巨大的能量波朝着三人飞去'
    '轰隆一声巨响'
    scene bg 3
    with Punch
    $renpy.pause(delay=0.2,hard=True)
    #ff14音乐3.1开始
    play music '<from 0 to 66>audio/sprechstimme.mp3' fadein 1.5
    $renpy.say(Character('·大&二&小·',color=c),'竟然没事？')
    '三人抬头，发现米娅举起魔杖挡在了三人前方'
    show cg 3:
        yoffset -100
    with Punch
    miya '夜之女王，你果然还是错了'
    miya '虽然她们三个都有缺点\n但是她们也都有不可替代的优点'
    miya '笨拙而冲动的二二，善于助人'
    miya '小小虽然懒散，但是她喜欢思考'
    miya '大大虽然软弱，但是她很努力\n而且...而且很大！'
    dada '很大是什么啦！'
    ayana '米娅没想到你现在只会说些一面之词了\n话稍微说多了，现在你们都直接消失吧！'
    hide cg
    with fade

    show cg 2:
        yoffset -100
    with Punch
    miya '可恶...\n看来三段LB也抵挡不住'
    miya '你们三个快走吧\n这里交给我就好'
    miya '到时候，开心地和兔拨鼠们一起玩吧'
    miya '我先找剧组领盒饭去了'
    hide cg
    $renpy.say(Character('·大&二&小·',color=c),'把你丢下就太残忍了啊')
    $renpy.say(Character('·大&二&小·',color=c),'MIYA！！')

    show cg 4:
        yoffset -100
    with fade
    miya '诶？'
    erer '谢谢你夸我！\n你的冰淇淋我还没吃够！\n怎么可能离开你！'
    dada '谢谢你！\n一直以来的努力我都没有木大！\n我会继续努力的！'
    xiaoxiao '我这么懒还真是对不起了\n等回去了我绝对不放过你！'
    ayana '力量增强了？'
    ayana '我怎么可能在这里倒下！\n全炮门，开火！'
    hide cg 4
    #音效
    play sound ayanacoming
    show cg 5:
        yoffset -100
    with Punch
    miya '可恶！还差一点点！！'
    ayana '把你们全——部——鲨——光——！！'
    miya '还有一点！！！'
    ayana '你们的力量增幅也就到此为止了么？一下子全灭你们！'
    #音乐结束
    stop music
    play sound chargeup
    'ayana，只是你没有看到，'
    #ff14音乐3.2开始
    play music '<from 68>audio/sprechstimme.mp3' fadein 1.5
    '在我们背后....'
    '即便不再有叫做{rb}奈能量{/rb}{rt}Akashic{/rt}的东西'
    '支撑着我们前行的——'
    '还有兔拨鼠的力量！'

    ##下面的充能环节使用点击按钮充能QTE
    ##添加一个充能条(bar),以显示充能进度
    $renpy.say(Character('QTE-Loading',color=c),'点击充能QTE')
    show screen tbar
    with dissolve
    $renpy.pause(delay=9.8,hard=True)
    $renpy.say(Character('QTE准备中',color=c),'点击释放QTE')
    dada '充能完成！\n{rb}百万核爆！！{/rb}{rt}Bombarda Maxima{/rt}'
    play sound ayanacoming
    hide screen tbar
    with dissolve
    miya '？！'
    miya '为什么还是赢不了？'
    miya '还是不行么？'
    xiaoxiao '是不是因为咱们在左边的缘故'
    dada '米娅你别多想了'
    erer '再努力一下'

    miya  '有道理！那我们换一下！'
    hide cg 5
    show cg 6:
        yoffset -100
    with fade
    ayana '嗯？'
    ayana '力量增强了？'
    ayana '可恶！'
    show cg 6:
        yoffset -100
    with Punch
    ayana '啊啊啊啊啊啊啊啊啊啊啊'
    stop music fadeout 2.0

    scene bg 1
    with Light
    #main音乐开始
    play music main fadeout 1.5
    show tuntun at ltl
    show nana normal at rc
    show miya at rtr
    tuntun '从那以后，大大、二二、小小变回了现在的你，这段故事也到此结束'
    nana '原来还有这种事...'
    miya '总之回来了就好~'
    miya '见到了各种各样的奈奈真的也挺高兴的'
    miya '虽然很累就是了'
    nana '真是对不起了米娅酱~'
    miya '以后要好好的\n不要想太多'
    miya '大家只想要陪伴在你身边\n不要把自己逼得太紧了哦'
    nana happy '嗯！今后也一起玩吧！'
    miya '话说你为什么还在这'
    tuntun '嘛~最后总要升华一下主题嘛\n所以还是想和miya以及兔拨鼠们一起说'
    miya '明白了~那...'
    miya '1'
    tuntun '2'
    $renpy.say(Character('·兔拨鼠·',color=c),'3')
    pause 1.5
    scene bg 1
    show tuntun at ltl
    show nana normal at rc
    show miya at rtr
    with Flash
    $renpy.say(Character('·MIYA&吞吞&兔拨鼠·',color=c),'{size=+10}生日快乐!{/size}')
    '祝奈奈永远健康！！'
    #$ renpy.movie_cutscene("over.webm") #此处接入片尾.....啊这片尾得自己搞
    pause 2.0
    scene developer
    developer '滋-滋——(假装是电流声)'
    developer '以下是来自developer的留言'
    developer '感谢游玩兔子地狱-RabbitInferno'
    developer '此为Beta版'
    developer '按照tuntun的剧本改变而成'
    developer '在主界面的<关于>中可以找到反馈的途径'
    developer '当然你直接去奈群或者里找<奈奈永远打不出来的喷嚏>也是没有问题啦~'
    developer '——'
    developer '以前没用过renpy编辑器写过游戏'
    developer '(甚至我就没写过类似的galgame)'
    developer '所以这算是我入门renpy的第一次练手'
    developer '由于国内很少用renpy写东西，所以现成资源很少啦，完全不够我copy(笑)'
    developer '所以这几天才爆肝啦~'
    developer '本来下定决心学个pr给游戏做片头片尾'
    developer '顺便把tuntun画的3只奈奈的gif用上'
    developer '但是写到这里才发现...'
    developer '离假期结束还有6.0416天'
    developer '不得不说我还有3张卷子和一本练习册....'
    developer '而且好巧不巧'
    developer '和奈奈一样，我也有胃病\n正好25号复查，照一下胃镜'
    developer '所以时间完全不够写作业啦！(高兴)'
    developer '但写作业哪有csgo好玩呢？(确信)'
    developer '那就开学再补吧~'
    developer '于是——'
    developer '游戏结束啦'
    developer '再次地，感谢游玩兔子地狱-RabbitInferno'





    #游戏结束
    return






label reply(code):
    #menu1
    if(code==0):
        tuntun '因为miya和nana贴贴！'
        miya '其实我..我不是很在意这点啦....\n哎呀关键绝对不是这个！'
        return
    elif(code==1):
        return

    #menu2
    elif(code==2):
        miya '你这是复刻了舔刀梗么？(无语)'
        miya '你就是比起世界和平，明天的口粮最重要的辣个蓝人？'
        erer angry '被发现了啊~sodayo(即答)'
        erer angry '可恶！'
        return
    elif(code==3):
        miya '怎么我还没动手你就倒了呀..'
        return

    #menu3
    elif(code==4):
        miya '五乘八等于？'
        xiaoxiao '40！你以为我的数学老师和某只豹子是同一个么'
    elif(code==5):
        miya '那你一定很平吧~'
        xiaoxiao '哈？！你以为我和某个魔法少女共用一个平板么'
        miya '啊确实，某只豹子也这么说，然后被红烧了..'
        xiaoxiao '那锅汤确实整挺香的'
        xiaoxiao '.....'
        xiaoxiao '？！(害怕)'
        xiaoxiao '哼！'
